

Assignment 2 Final Report of HasWasm: A WebAssembly-like Embedded DSL in Haskell

Faiz Ilham Muhammad (5231981) and Xinliang Lu (0822760)

30 June 2023

1 Introduction

HasWasm is a domain-specific language (DSL) that is integrated into Haskell. It makes use of the strong type system in Haskell to offer expressive and type-safe abstractions for writing WebAssembly code. HasWasm supports two basic types, blocks and control flows, functions, local and global variables, and export and imports. It also includes a code generator that turns the parsed Abstract Syntax Tree (AST) into a WebAssembly module in text (WAT) format.

This project is aimed to create a type-safe, expressive embedded DSL for WebAssembly programming in Haskell. Our effort is concentrated on offering expressive, type-safe, and user-friendly language abstractions for writing WebAssembly code.

2 Approach

2.1 HasWasm Overall Design

In designing the type signatures, we employ type families to define relationships between each types. We also want to avoid using any undecidable language extensions (e.g. UndecidableInstances) so that the compile time can be guaranteed to be terminating.

In HasWasm, a WebAssembly instruction's type is represented as a *TypedInstr*. For example, instruction $i32.add :: i32\ i32 \rightarrow i32$ is typed as follows.

```
1 i32_add :: (Stack s) => TypedInstr (s :+ I32 :+ I32) (s :+ I32)
```

This means that `i32_add` is an instruction that requires a `Stack s` with two `I32` value on top of it, and will results in `Stack s` with only a single `I32` value on top of it. Sequencing the instructions can be done easily using our defined `#` operator.

```
1 (#) :: (Stack a, Stack b, Stack c) => TypedInstr a b -> TypedInstr b c -> TypedInstr a c
```

A function is represented by the `WasmFunc` type:

```
1 myFunc :: WasmFunc p v r
```

in which `p` is the types of parameters, `v` is the types of local variables, and `r` is the types of returned values. Each of `p`, `v` and `r` can be a unit type, an `I32`, a `F32`, or tuple of `I32` and/or `F32` (up to 5 tuple members). Those types are grouped in a type family called `VarTypes`.

```
1 class BaseType t where ...
2 instance BaseType I32 where ...
3 instance BaseType F32 where ...
4
5 class VarTypes v where ...
6 instance VarTypes () where ...
7 instance VarTypes I32 where ...
8 instance VarTypes F32 where ...
9 instance (BaseType t1, BaseType t2) => VarTypes (t1, t2) where ...
10 instance (BaseType t1, BaseType t2, BaseType t3) => VarTypes (t1, t2, t3) where ...
11 ...
```

To declare a function, users may use `createLocalFunction` or `createExpFunction`, which take a string for function name, and a function that takes a `VarSet` of parameters, a `VarSet` of local variables, a return instruction and returns the function body. The return instruction is provided by the `createFunction` so that it have the correct typing.

```

1 createExpFunction :: (VarTypes p, VarTypes v, VarTypes r) =>
2   String -> (VarSet p -> VarSet v -> ReturnInstr (StackT s t) r -> FuncBody (StackT s t)
   r) -> WasmFunc p v r

```

In practice, users can declare a function as shown below.

```

1 -- takes 3 i32 parameter, have 1 i32 local, and return 1 i32
2 add :: WasmFunc (I32, I32) (I32) I32
3 add = createLocalFunction "rgb" func
4   where
5     -- Note: func type signature can be omitted for simplicity
6     func :: (Stack s) => (Var I32, Var I32) -> Var I32 -> ReturnInstr s I32 -> FuncBody s
       I32
7     func (a, b) l ret =
8       local_get a #
9       local_get b #
10      i32_add #
11      ret           -- no need explicit return here, but it can also use one

```

Blocks and loops use similar type family in their signatures. It takes a VarTypes of prerequisite types p, a VarTypes of resulting types r, and a function that takes the label of the block or loop, and returns the block body. StackType r s is equivalent to a type t_r^* , and FuncCallType s p r is equivalent to $t_p^* \rightarrow t_r^*$.

```

1 block :: (VarTypes p, VarTypes r, Stack s) => p -> r -> (TypedLabel (StackType r s) ->
   FuncCallType s p r) -> FuncCallType s p r
2
3 loop :: (VarTypes p, VarTypes r, Stack s) => p -> r -> (TypedLabel (StackType p s) ->
   FuncCallType s p r) -> FuncCallType s p r

```

In practice, users can use block, loops, and branching as such:

```

1 factorial :: WasmFunc I32 () I32
2 factorial = createExpFunction "factorial" func
3   where
4     func n () ret =
5       i32_const 1 #
6       block (I32) (I32) (\end-> -- end is the label id of the block
7         loop (I32) (I32) (\start -> -- start is the label id of the loop
8           local_get n #
9           i32_const 1 #
10          i32_le_s #
11          br_if end #      -- if n <= 1, jump to end
12          local_get n #
13          i32_mul #        -- total = total * n
14          local_get n #
15          i32_const 1 #
16          i32_sub #
17          local_set n #    -- set n = n - 1
18          br start        -- loop back
19        )
20      )

```

Build Implicitly Called Funcs and Global Vars By using the breadth-first search (BFS), this module recursively detects and automatically checks the following properties for a function or global variable:

- Being defined twice with different type signatures
- Being called recursively and implicitly

Then automatically adding the missing declarations for these functions and variables. Also, this module can prevent infinite nested calls.

2.2 Supported Wasm Instructions

HasWasm supports both basic i32 and f32 instructions, these include the following calculation operations¹:

- $unop_{f32}$: neg, abs, ceil, floor, trunc, nearest, sqrt

¹<https://dl.acm.org/doi/pdf/10.1145/3062341.3062363>

- *binop_{i32}*: add, sub, mul, div_s, div_u, rem_s, rem_u, and, or, xor, shl, shr_s, shr_u, rotl, rotr
- *binop_{f32}*: add, sub, mul, div, min, max, copysign
- *testop_{i32}*: eqz
- *relop_{i32}*: eq, ne, lt_s, lt_u, gt_s, gt_u, le_s, le_u, ge_s, ge_u
- *relop_{f32}*: eq, ne, lt, gt, le, ge

and other instructions¹:

- *i32.const*, *f32.const*
- block *tf e** end
- loop *tf e** end
- br *i*, br_if *i*
- call *i*, return
- get_local *i*, set_local *i*
- get_global *i*, set_global *i*

3 Results and Discussion

In this section, we will show how to use HasWasm and discuss its features.

3.1 Accomplished Functionalities

- Utilizing type families and Generalized Algebraic Data Types (GADT) to guarantee type-safety.
- Generating WebAssembly code in text format.
- Supporting basic i32 & f32 instructions, blocks and control flows (if-else, jumps), making and declaring functions & local variables & global variables, exporting and importing functions.

3.2 Example

Here we provide an example for using HasWasm. Users first declare exported functions using “createExpFunction”, or local functions using “createLocalFunction”:

```

1  -- an exported factorial function, that takes an i32 and returns i32
2  factorial :: WasmFunc I32 () I32
3  factorial = createExpFunction "factorial" func
4      where
5      func n () ret =
6          i32_const 1 #
7          block (I32) (I32) (\end->
8              loop (I32) (I32) (\start ->
9                  local_get n #
10                 i32_const 1 #
11                 i32_le_s #
12                 br_if end #      -- if n <= 1, jump to end
13                 local_get n #
14                 i32_mul #        -- total = total * n
15                 local_get n #
16                 i32_const 1 #
17                 i32_sub #
18                 local_set n #    -- set n = n - 1
19                 br start        -- loop back
20             )
21         )
22
23 -- recursive version of the factorial function called "factorial_rec"
24 -- it is also exported and takes an i32 and returns i32
25 factorialRec :: WasmFunc I32 () I32

```

```

26 factorialRec = createExpFunction "factorial_rec" func
27   where
28     func n _ ret =
29       block () () (\lbl ->
30         local_get n #
31         br_if lbl #
32         i32_const 1 #
33         ret          -- return 1 if n == 0
34       ) #
35       local_get n #
36       local_get n #
37       i32_const 1 #
38       i32_sub #
39       call factorialRec #
40       i32_mul
41
42 -- an exported function that increase a global "counter" variable
43 incCounter :: WasmFunc () () ()
44 incCounter = createExpFunction "inc_counter" func
45   where
46     func _ _ _ =
47       global_get counter #
48       call addByConst #
49       global_set counter #
50       global_get counter #
51       call consoleLog
52
53 -- a local function that increase the given argument
54 -- with some value in the "constant" global variable
55 addByConst :: WasmFunc I32 () I32
56 addByConst = createLocalFunction "addByConst" func
57   where
58     func i _ _ =
59       local_get i #
60       global_get constant #
61       i32_add

```

Users can also declare global variables and imported functions:

```

1 -- a mutable global variable of type i32
2 counter :: GlobalVar Mut I32
3 counter = createGlobalI32 "counter" (Just "counter") 0
4
5 -- an immutable global variable of type i32
6 constant :: GlobalVar Imm I32
7 constant = createGlobalI32 "constant" Nothing 1
8
9 -- an import function named "console_log" that takes i32 and returns nothing
10 -- the function should be provided with console.log by the host of the wasm code
11 consoleLog :: WasmFunc I32 () ()
12 consoleLog = createImportFunction "console" "log" "console_log"

```

Finally, users can make a module that includes all the declared functions and global variables. Functions or global variables that are not added manually by the user, but are called indirectly by one of the included functions will also be included automatically into the module. In this example, “constant”, “counter”, “addByConst”, and “consoleLog” will be automatically included because “incCounter” directly or indirectly uses them.

```

1 myModule :: WasmModule
2 myModule = createModule $ do
3   addFunc factorial
4   addFunc factorialRec
5   addFunc incCounter

```

Taken together, the full program would look like this:

```

1 module Main where
2
3 import HasWasm
4 import HasWasm.Instruction
5
6 main :: IO ()
7 main = do
8   case buildModule myModule of

```

```

9     Right result -> putStrLn $ result
10    Left err -> putStrLn $ "Error: " ++ err
11
12 myModule :: WasmModule
13 myModule = createModule $ do
14     addFunc factorial
15     addFunc factorialRec
16     addFunc incCounter
17
18 -- an exported factorial function, that takes an i32 and returns i32
19 factorial :: WasmFunc I32 () I32
20 factorial = createExpFunction "factorial" func
21     where
22         func n () ret =
23             i32_const 1 #
24             block (I32) (I32) (\end->
25                 loop (I32) (I32) (\start ->
26                     local_get n #
27                     i32_const 1 #
28                     i32_le_s #
29                     br_if end #      -- if n <= 1, jump to end
30                     local_get n #
31                     i32_mul #        -- total = total * n
32                     local_get n #
33                     i32_const 1 #
34                     i32_sub #
35                     local_set n #    -- set n = n - 1
36                     br start        -- loop back
37                 )
38             )
39
40 -- recursive version of the factorial function called "factorial_rec"
41 -- it is also exported and takes an i32 and returns i32
42 factorialRec :: WasmFunc I32 () I32
43 factorialRec = createExpFunction "factorial_rec" func
44     where
45         func n _ ret =
46             block () () (\lbl ->
47                 local_get n #
48                 br_if lbl #
49                 i32_const 1 #
50                 ret          -- return 1 if n == 0
51             ) #
52             local_get n #
53             local_get n #
54             i32_const 1 #
55             i32_sub #
56             call factorialRec #
57             i32_mul
58
59 -- an exported function that increase a global "counter" variable
60 incCounter :: WasmFunc () () ()
61 incCounter = createExpFunction "inc_counter" func
62     where
63         func _ _ _ =
64             global_get counter #
65             call addByConst #
66             global_set counter #
67             global_get counter #
68             call consoleLog
69
70 -- a local function that increase the given argument
71 -- with some value in the "constant" global variable
72 addByConst :: WasmFunc I32 () I32
73 addByConst = createLocalFunction "addByConst" func
74     where
75         func i _ _ =
76             local_get i #
77             global_get constant #
78             i32_add
79
80 -- a mutable global variable of type i32
81 counter :: GlobalVar Mut I32

```

```

82 counter = createGlobalI32 "counter" (Just "counter") 0
83
84 -- an immutable global variable of type i32
85 constant :: GlobalVar Imm I32
86 constant = createGlobalI32 "constant" Nothing 1
87
88 -- an import function named "console_log" that takes i32 and returns nothing
89 -- the function should be provided with console.log by the host of the wasm code
90 consoleLog :: WasmFunc I32 () ()
91 consoleLog = createImportFunction "console" "log" "console_log"

```

When it runs, the program will produce the following WebAssembly code in text format:

```

1 (module
2   (import "console" "log" (func $console_log (param i32) ))
3   (export "counter" (global $counter))
4   (export "factorial" (func $factorial))
5   (export "factorial_rec" (func $factorial_rec))
6   (export "inc_counter" (func $inc_counter))
7   (func $addByConst (param i32) (result i32)
8     local.get 0
9     global.get $constant
10    i32.add
11  )
12  (global $constant i32 (i32.const 1))
13  (global $counter (mut i32) (i32.const 0))
14  (func $factorial (param i32) (result i32)
15    i32.const 1
16    (block $l0 (param i32) (result i32)
17      (loop $l1 (param i32) (result i32)
18        local.get 0
19        i32.const 1
20        i32.le_s
21        br_if $l0
22        local.get 0
23        i32.mul
24        local.get 0
25        i32.const 1
26        i32.sub
27        local.set 0
28        br $l1
29      )
30    )
31  )
32  (func $factorial_rec (param i32) (result i32)
33    (block $l0
34      local.get 0
35      br_if $l0
36      i32.const 1
37      return
38    )
39    local.get 0
40    local.get 0
41    i32.const 1
42    i32.sub
43    call $factorial_rec
44    i32.mul
45  )
46  (func $inc_counter
47    global.get $counter
48    call $addByConst
49    global.set $counter
50    global.get $counter
51    call $console_log
52  )
53 )

```

3.3 HasWasm Evaluation

We evaluate HasWasm by answering the following questions:

Does the DSL ensure the type correctness of the constructed Wasm code? The answer is yes, the sequencing operator ($\#$) and the type families for functions and block ensures that the generated WASM code will be type safe. However, our types are currently more strict than the WebAssembly specification, in particular for `br` and `return`. For example the specification defines that, for $C(i) = t^*$, the typing judgment of `br` is $C \vdash \text{br } i : t_1^* t^* \rightarrow t_2^*$, while our implementation of `br` instruction has the equivalent of typing judgment $C \vdash \text{br } i : t^* \rightarrow t_2^*$.

Is the syntax similar enough to wasm text syntax? The answer is yes, in the above code example, we show that HasWasm provides a similar way to directly manipulate the stack like WebAssembly but using GADT to ensure the type-safety.

4 Limitations

There are several limitations in our project that we couldn't implement due to limited time:

1. Functions can only have at maximum 5 parameters, 5 local variables, and 5 return values. The same limitations also apply for block and loop prerequisite and resulting type.
2. The type signature for `br` and `return` is more restrictive than the specification, meaning that there are programs that should be allowed but will be rejected by our DSL.
3. Name clash checking of functions and global variables are done in runtime, not compile time.
4. Code optimizer and binary-code generator are not implemented.

5 Conclusion

We have developed HasWasm, an embedded DSL for writing WebAssembly code in Haskell. The project's primary goals, which included developing a type-safe and expressive tool for Wasm code generation, were met. Future development can be focused on fixing the limitations in our implementation.